Metal Commands: Section "Assorted and Compiler Related"

Commands that don't start with A through Z

The following section is things like *, ?, <>, etc. Inside of Metal, these are referred to as "Group Zero" commands (Group One is the ones that start with A, and Group Twenty-Six would be the ones (if there were any) that start with Z).

After these commands are the explanations on the Compiler Directives (.ESCLEAD, //, etc).

?

Type: command alias

  Is: PRINT

The question-mark ("?") is an optional short-hand way to say "PRINT". 99% of all Basic-type systems allow you to use this way to say Print without saying it.

See also: .., \, \$, ', ", |, `, <<, >>, +=, -=, /=, *=, PRINT

..

Type: command alias

  Is: TO

Two periods in a row is an optional way to say "TO". While this is taking up the same amount of space (two characters), again, the majority of Basic compilers understand the two are the same.

See also: ?, \, \$, ', ", |, `, <<, >>, +=, -=, /=, *=, TO

\

Type: command alias

  Is: CR

This is an optional short-hand way to say "CR". Languages similar to Metal use this, so once again, it allows it to be used.

See also: ?, .., \$, ', ", |, `, <<, >>, +=, -=, /=, *=, CR

\$

Type: command alias

  Is: CR$

A logical extension of the \ - does the same thing as CR$.

See also: ?, .., \, ', ", |, `, <<, >>, +=, -=, /=, *=, CR$,

:

Type: command breaker

print "Hi!":goto Bye

Most of any Metal program is going to be colons. They separate commands from each other, allowing both logical and visual cues on how the system will operate.

In some instances, colons are optional. For example:

  if a=1 : print "One"

Here the colon is optional, since Metal is smart enough to realize that the Print command is not a math function, so it stops after the a=1.

However, other times the colon is required:

  if a=1 print "One":goto OneIsTheNumber

If the colon was missing in this example, Metal would bomb out after printing the word "One" with a Syntax Error - this is because the Print command doesn't expect to see a Goto in the middle of it's data; it needs to know where to stop.

See also:  ,  THEN

+

Type: math/string operand

a = b + c

a$ = b$ + c$

The plus sign simply means "add the previous number to the next number". It is also used to join strings together.

See also: -, *, /, MOD

-

Type: math operand

a = b - c

The minus sign will subtract the next number from the previous number. In the example above, "c" will be taken away from "b" and placed in "a".

Subtraction can only be performed on numbers, not strings.

See also: +, *, /, MOD

*

Type: math operand/buried command

a = b * c

Print b$*c

The asterisk is used to signify multiplication. It can also be used in the Print command for repeating strings.

When used with numbers, the asterisk means to multiply the first number by the second and place the result back.

When used in the Print command, it will repeat the given string (b$ in our example) for the given number of times (c in our example). The following line will print out "Tora! Tora! Tora!":

  print "Tora!"*3

See also: +, -, /, MOD, PRINT

/

Type: math operand

a = b / c

The forward slash (or just slash) is used to divide the first number by the second. Since Metal operates with only integer numbers, the result will never be a fraction. For example:

          5 / 2

Will result in a value of 2, not 2.5

See also: +, -, *, MOD

==

Type: command

a == +5

The double equals sign is used by Metal to signify that you plan to perform the operation on the current value. In other words, the two following lines are the same:

       a = a+5

       a == +5

The two lines will work 100% the same, but the ==+5 version will run slightly faster, since it doesn't have to re-gather the variable data for "a" twice.

The double equals can only be used when assigning the value of a variable, never in a math expression.

See also: =, +=, -=, *=, /=

+=

Type: command alias

  Is: ==+

The plus-equals combination is translated by Metal into ==+. Thus the following three lines are all the same:

          a = a+5

          a +=  5

          a ==+ 5

See also: ==, =, -=, *=, /=, \$, ?

-=

Type: command alias

  Is: ==-

The minus-equals combination is translated by Metal into ==-. Thus the following three lines are all the same:

          a = a-5

          a -=  5

          a ==- 5

See also: ==, =, +=, *=, /=, \$, ?

*=

Type: command alias

  Is: ==*

The times-equals combination is translated by Metal into ==*. Thus the following three lines are all the same:

          a = a*5

          a *=  5

          a ==* 5

See also: ==, =, +=, -=, /=, \$, ?

/=

Type: command alias

  Is: ==/

The divide-equals combination is translated by Metal into ==/. Thus the following three lines are all the same:

          a = a/5

          a /=  5

          a ==/ 5

See also: ==, =, -=, +=, *=, \$, ?

(

Type: command delimiter

The left parenthesis is used to mark the start of either a data set (such as LEN(s$) ) or a math sub-expression (such as a=5*(b+2) ). It must be balanced with a right parenthesis, otherwise Metal will get confused.

See also: )

)

Type: command delimiter

This is the right-side balance of the left parenthesis. You must have this in order for Metal to understand what you mean in most commands and math expressions.

See also: (

,

Type: buried command

The comma is used to separate data sets inside of commands. For example:

       open #1,f$

The comma here is used to tell Metal that it holds between the file channel (#1) and the filename (f$).

See also: :,  THEN

[

Type: buried command

The left-square bracket is used in some commands to flag subsets of data groups, or special exceptions. It is also used when reading and setting variable strings to allow access to sub-parts of strings.

See also: ], INPUT, Variable Usage

]

Type: buried command

This is the balance for the left-square bracket.

See also: [, INPUT, Variable Usage

#

Type: buried command

The pound sign (number sign to some people) is used by Metal to flag that a file channel is to be used.

See also: CLOSE, INPUT, FCOPY, FILEINFO, FILEMARK, FILESIZE, FILEAUXTYPE, FILETYPE, OPEN, POSITION, PRINT, TCOPY

;

Type: buried command

The semi-colon is used by the PRINT command to separate expressions, and to force the it to not print a carriage return at the end of an expression list.

See also: PRINT

"  '  |  `

Type: text block delimiters

Metal supports four single-character text block delimiters. Once the Metal Compiler finds one of the four, it assumes that the source code following is to be used as literal text until another of the exact same one is found. Thus, you can do the following:

          print "'Text'"

          print '"Text"'

          print |"'Text'"|

          print `"'|Text|'"`

And Metal will output the following:

'Text'

"Text"

"'Text'"

"'|Text|'"

If you need to bury the same text delimiter character that you are using (say for example, the single quote), then you must use the Escape Character followed by the delimiter (example: \' in the text will put a ' in it). This requires that the .ESCLEAD directive is in use.

See also: <<, >>, .ESCLEAD

<<

Type: text block delimiter, start

The double less-then is used by Metal to specify a text block where you plan to use most of the text delimiters. You end a block with such a lead-in with >> (double greater-then).

See also: >>

>>

Type: text block delimiter, end

End of a text block that was started with <<.

See also: <<

<>

Type: logic operand

result = value1 <> value2

a = b <> 5

if a$<>"END" goto More

The combination of the less-than sign with the greater-than sign results in what Basic-like programs (which Metal is) call a "not-equals-to" representation.

The not-equals can be performed on either numerical or string values.

If numerical values are being compared, if they are not the same values (for example if B is not equal to 5), then the not-equals operation will return a value of 1. Other wise, if B is equal to 5, then the operation will return a zero.

If strings are being compared, they must match in both length, text, and case. Thus "Bonnie" is not equal to "BONNIE", and "Clyde" is not equal to "John".

See also: =, <, >, <=, =>, NOT, OR, AND, EOR

><

Type: logic operand, alias

  Is: <>

This is an alias for <>. Metal treats both >< and <> as the same operation (in fact it converts >< into <> for you).

See also: <>

<=

Type: logic operand

result = value1 <= value2

a = b <= 5

if nl <= 30 goto Small

if a$ <= "a" goto NotLower

By combining the less-than with the equals sign, you get an "less than or equals to" operation.

This operand will return a value of 1 if the value1 value is less than or equal to the value of value2. Thus in our first example, A will be zero 0 if b is greater than 5, but will be 1 if less than or equal to 5.

You can also perform this operation on strings. Metal compares strings from front to back, stopping when it runs out of characters on either string. If the ascii code of a character in value1 is greater than the ascii code of a character in the same position in value2, the operation will return a zero value.

See also: =, <, >, =>, <>, NOT, OR, AND, EOR

=<

Type: logic operand, alias

  Is: <=

This is an alternate form for the <= operand. The Metal Compiler will actually translate this into the <= for you, and a Trace will display this as "<=".

See also: <=

=>

Type: logic operand

result = value1 => value2

a = b => 5

if nl => 30 goto Big

if a$ => "a" goto IsLower

By combining the equals sign with the greater-than sign, you get an "greater than or equals to" operation.

This operand will return a value of 1 if the value1 value is greater than or equal to the value of value2. Thus in our first example, A will be zero 0 if b is less than 5, but will be 1 if greater than or equal to 5.

You can also perform this operation on strings. Metal compares strings from front to back, stopping when it runs out of characters on either string. If the ascii code of a character in value1 is less than the ascii code of a character in the same position in value2, the operation will return a zero value.

See also: =, <, >, <=, <>, NOT, OR, AND, EOR

>=

Type: logic operand, alias

  Is: =>

This is an alternate form for the => operand. The Metal Compiler will actually translate this into the => for you, and a Trace will display this as "=>".

See also: =>

=

Type: logic operand/command

result = value1 = value2

variable = value

if b=5 goto ItIsFive

a=7

The equals sign is the most commonly used operand in Metal. It is used in two forms: to set a variable to a value (LIMIT=7*age), and to check the value of some expression (if user=1 goto SysOp).

When the equals sign is used to set a variable to some value, we are said to "equate" the variable to the given value (or result of the expression - value does not have to be a number. It can be something like FileSize(1), for example). By using this we set the variable to the given value. This is the only "real" way that variables can be changed (the "==" command is actually the "=" sign with a special function attached, and externals and system events can change variable values, but these are special cases).

When the equals sign is used either inside of an math expression or in an If-Then command, it is said to "compare" the values, returning a value of zero if the results of value1 and value2 are not equal. If the two values are exactly the same (both are 7, for example), then the equals sign operation will return a 1.

The equals sign also compares strings. Case is important with the equals sign, as is the length of the strings. Thus "a" is equal to "a", but "AA" is not equal to "A", and "a" is not equal to "A".

See also: ==, <, >, <=, <>, +=, -=, /=, *=, NOT, OR, AND, EOR

<

Type: logic operand

result = value1 < value2

a = b < 5

if nl < 30 goto Small

if a$ < "a" goto NotLower

The less-than sign is just that - it informs Metal to return a value of 1 if value1 is less than value2. If value1 is greater than or equal to value2, it returns a 0.

This operation also works on strings. Metal compares strings from front to back, stopping when it runs out of characters on either string. If the ascii code of a character in value1 is greater than or equal to than the ascii code of a character in the same position in value2, the operation will return a zero value.

See also: =, >, =>, <=, <>, NOT, OR, AND, EOR

>

Type: logic operand

result = value1 > value2

a = b > 5

if nl > 30 goto Big

if a$ > "Z" goto IsLower

The greater-than sign is just that - it informs Metal to return a value of 1 if value1 is greater than value2. If value1 is less than or equal to value2, it returns a 0.

This operation also works on strings. Metal compares strings from front to back, stopping when it runs out of characters on either string. If the ascii code of a character in value1 is less than or equal to than the ascii code of a character in the same position in value2, the operation will return a zero value.

See also: =, <, =>, <=, <>, NOT, OR, AND, EOR

Compiler Directives

The following section is where we describe all the rest of the commands that Metal recognizes. This section does not pertain to how Metal understands and uses the commands you give it; it tells you, instead, what the Compiler does when you give it some text to act on.

All directives that start with a period (.ALLGLOBAL, .CHAINFILE, etc) must be the first thing on a line. In other words, if you indent the .ALLGLOBAL by even one space, the Compiler will not know what you are trying to do.

.ALLGLOBAL

Type: compiler directive

.ALLGLOBAL

The AllGlobal directive tells the Compiler that all labels in the source file are to be considered global (this excludes local labels, though). Doing this allows other programs to enter the current source file at any label point, and also has the side effect of making the runtime object file larger (since it has to store the label names and addresses off).

See also: GLOBAL, RUN, RUNSUB

.CHAINFILE

Type: compiler directive

.CHAINFILE filename

.CHAINFILE 1/moses.p2

The .CHAINFILE is similar to the .INCLUDE directive, but it doesn't remember where you came from - it switches from the current source file to the sourcefile in the filename parameter. This filename must be fully qualified, unless you are using the Scriptor Command DefIncPath or have used a .PREFIX command prior to this.

The compiled runtime object file remains the same as if you didn't use the .CHAINFILE directive, but any text after the line is ignored, since the current source file is closed and the new one opened and read in its place.

If you use .CHAINFILE inside of an .INCLUDEFILE, the Compiler will chain the Include files together.

See also: .INCLUDE, .PREFIX, DEFINCPATH

.DEFMODE

Type: compiler directive

.DEFMODE type value

.DEFMODE get 1

.DEFMODE fcopy 4

The .DEFMODE (DEFault MODE) directive tells the Compiler (and Metal, when it runs the object file) to override the default values for some of the commands. The command you wish to override is given in the type field, with the new default (for this object file only) value given.

type may be one of the following strings:

          GET       this will be the default for a GET operation

          FCOPY     this will be the default Fcopy mode

          INPUT     this will be the default for INPUTing from the keyboard

You must refer to the documentation on Get, Fcopy, and Input for the default values and the possible value ranges.

See also: GET, FCOPY, INPUT, MODE

.DEFSTR

Type: compiler directive

.DEFSTR~name~ = string

.DEFSTR~Lives~ = 5

The .DEFSTR (DEFine STRing) is a way to let you put "magic numbers" (values that the program depends on, uses a lot, and changing all fifty of them is a pain) or strings into a referable name that can be then changed in one place rather than fifty or sixty places.

The name must start and end with the ~ (tilde) character, followed by a space, then an equal sign, space, and the string. The format is important, as the Compiler will get very confused if you don't follow this format!

Using the defined name later in the source is as easy as typing it again - if you have "a=~Lives~", the Compiler will see that you set "~Lives~" to five and it puts the string into it.

You can also use the name inside of text blocks by using the EscLead character (usually a reverse backslash). For example: print "The max lives is \~Lives~." will display as "The max lives is 5."

Nesting DEFSTRings together can be done, but is not recommended.

Recursive nesting of strings (having ~Lives~ refer back to itself in some way or another) will result in the Compiler appearing to "lock up" until the object file expands to 16megs or the system runs out of disk space, whichever comes first.

.END

Type: compiler directive

.END

Normally, the Compiler doesn't need this command, since it will know when it reaches the end of the source file(s), but you may wish to place this in before the end to stop the Compiler prematurely. It is entirely optional.

.ESCLEAD

Type: compiler directive

.ESCLEAD="char"

.ESCLEAD="\"

By default, the Compiler doesn't use an ESCape LEAD-in character, since it is foreign to most Apple programmers. However, Unix and C programmers are quite used to it, allowing them to bury special characters in their text files without resorting to putting literal control characters in it.

The Compiler will not let you set the Escape Lead character to any of the following: control characters, spaces, ', ", |, or `. Normally, you would use the reverse back-slash, since this is commonly used form of this.

Assuming the Escape Lead character is the backslash, the following table can be used:

       text         means...

       ----         --------------

       \I           inverse (control-O embedded in object code)

       \M           mousetext (control-P embedded in object code)

       \P           plain (aka normal) (control-N embedded in object code)

       \Ccr       repeat character c for r times (or rr or rrr times)

                    (control-R, c, r embedded in object code)

       \Gx,y        goto x,y position (or xx or xxx or yy or yyy)

                    (control-^, asciicode 'x+32', asciicode 'y+32' in object)

       \A           bell (control-G inserted)

       \N           newline (control-M, <NOT> the control-F C uses)

       \R           cr (control-M, this is for C dead-heads)

       \L           line feed (control-F, this is the C newline)

       \T           tab (control-I)

       \B           backspace (control-H)

       \F           form feed; screen clear (control-L)

       \\           use the escape char (ex: ~~ or \\ will be ~ or \)

       \"           double quote

       \'           single quote

       \|           pipe           (these four are used to allow you to

       \`           reverse quote   bury quote marks in the file)

       \^c        control character c (ex: \^O does a control-O)

       \Xhex      hex value (ex: \XA0 is a ascii high space)

       \Dnum      decimal value (ex: \D160 is a ascii high space)

       \~name~    replacement Defined String text

See also: .DEFSTR

.INCLUDEFILE

Type: compiler directive

.INCLUDEFILE filename

.INCLUDEFILE 1/i/stnd.subs

.INCLUDEFILE network.manager

The .INCLUDEFILE directive lets the Compiler use additional source files without having to duplicate those files over and over for each program that needs them. This allows you to have "libraries" of Include Files, such as Standard.Routs or Pretty.Input, etc.

If the .INCLUDEFILE is not given a full pathname (1/i/stnd.subs in our first example), it will put the current .PREFIX or DefIncPath onto it.

Once the Included File is done with, the Compiler returns back to the first file at the next line after the .INCLUDEFILE

If you use .INCLUDEFILE inside of an Include, the Compiler will act as if you are Chaining the Includes together (in other words, it won't get confused).

See also: .CHAINFILE, .PREFIX, DEFINCPATH

.MINCOMP

Type: compiler directive

.MINCOMP major.minor

.MINCOMP 3.40

The .MINCOMP (MINimal COMPiler version) will check for the minimal revision of the Compiler that you want. If it doesn't find it it (say you have 3.38 instead of 3.40), the Compiler complains, and flags the object file as "wrong", which will force the system to re-compile it the next time it is used.

This directive is entirely optional.

See also: .MINVER

.MINVER

Type: compiler directive

.MINVER major.minor.sub

.MINVER 1.08.79

.MINVER 1.09

The .MINVER (MINimal VERsion) will check for the minimum version of the Metal Language. If you have program that require something that only appeared in later revisions (like the SWAP or TRAP XFER commands), then using this will have the same effect as .MINCOMP

This directive is also entirely optional.

See also: .MINCOMP

.PREFIX

Type: compiler directive

.PREFIX="pathname"

.PREFIX="1/i/"

The .PREFIX command allows to to set, when the source is being Compiled, the default prefix that ProDOS-8 will use. When the Compiler exits, the previous prefix value is restored.

By default, if the DEFINCPATH is set, the Compiler will use that string as the .PREFIX value upon entry. However, you may override this at any time by using this directive.

Again, this directive is optional.

See also: .CHAINFILE, .INCLUDE

.SAVEAS

Type: compiler directive

.SAVEAS "filename"

.SAVEAS "/HD2/GAMESCOMP/GBL.WAR.MAIN"

By default, Metal's Compiler will save the object file in either one of two places: in the same directory as the source file, or in the directory specified in COMPDIR in the Config file.

Some people like to have COMPDIR set to a RAM-disk to speed things up and keep wear-and-tear down on their system.

However, these RAM-based disks are not very large (generally 1-2 megs in size), and the object code of Future Vision alone is over a megabyte in space, so using any additional programs (such as Global War or Clash Of Arms) is impossible since you would run out of space quickly.

This is where the .SAVEAS comes into play.

Programs like games or voting or databases are not used much (message bases, email, and file transfers are the top ones). So why not plunk those down on the disk?

Simply set .SAVEAS to whatever complete filename you want to save the object code as and Metal will find it. All it costs is one (1) block on the normal place you put object code.

You don't have to modify any other programs, since Metal will "know" where the "real" program is at. You can even do multiple .SAVEAS's, one right after another, leaving a trail of 1-block files behind for Metal to follow.

The only restriction is that this must be the first line in the source file, since Metal has to know where to put things before it starts generating object code.

See also: COMPDIR

->

Type: compiler directive

The "text arrow" symbol is used by the Compiler to let it know that the line ends at this arrow and continues with the next one. Some editors (like Metal's built-in ones) can't handle more than a certain number of characters per line, and the Compiler can only handle lines in 250-byte chunks anyways.

For example:

          print "Hi! I am a Metal program!" ->

          "And I love to answer calls!"

The Compiler will see the above two lines as:

          print "Hi! I am a Metal program!""And I love to answer calls!"

If you want, you can "chain lines" together for most of your source code. Neither Metal nor the Compiler will care.

But beware, If-Then-Else commands rely on end of lines to know when to stop!

See also: !, //, (*, *), /*, */

!

Type: remark/compiler directive

! this is a remark

print "Hello" ! greetings, programs

The ! (exclamation point) is one of Metal's many ways of putting comments or "remarks" in your source code.

Commenting source code is good practice. The Compiler strips out comments as it goes, so putting a lot in your source will not slow down runtime nor compilation time. It will only make your source file big.

Anything from the ! to the end of the line is assumed to be a remark, and is ignored. The Compiler goes directly to the next line of code.

See also: //, /*, */, (*, *)

//

Type: remark/compiler directive, alias

  Is: !

// Another comment

print "Is this C++?"  // No, but sure looks like it.

Yes, we admit it, we bowed to pressure and allowed the Compiler to understand that the // (double-slash) is the same as a exclamation point. While the programmer of Metal doesn't care much for C or C++, he does realize that some people prefer it, and tries to make everyone happy.

See also: !, /*, */, (*, *)

/*

Type: remark block start/compiler directive

/* This is a really huge comment block.

   It covers two lines */

Pascal programmers are used to using "/* ...comment... */" as the way they put comments and remarks in their programs. Since Metal doesn't care, and the Compiler doesn't care, why should we?

The slash-asterisk starts the comment block, which goes until the Compiler files a */ at which point it stops ignoring source text and continues compiling code.

If the source file ends before the comment block does, the Compiler throws a fit in a very big way.

See also: !, //, */, (*, *)

*/

Type: remark block end/compiler directive

This ends the comment block that was started with /*.

See also: !, //, /*, (*, *)

(*

Type: remark block start/compiler directive, alias

  Is: /*

(* Hey, it's a C-style comment block! *)

Yes, Metal also allows you to use comment blocks starting with (*.

Using (* instead of /* means that you must end the block with *).

See also: !, //, /*, */, *)

*)

Type: remark block end/compiler directive, alias

  Is: */

What you need to use to end a comment block started with (*.

See also: !, //, /*, */, (*

label

Type: compiler directive/label definition

Enter

GameOver

Labels are how you tell Metal where things are at. Unlike AppleSoft, Metal doesn't use line numbers, it uses labels - text strings that are used to refer to parts of the program. For example, which one is easier to understand:

  AppleSoft Basic                            Metal Basic

  10 for j = 1 to 50                     for j = 1 to 50

  20 print j;                            print j:

  30 if j>10 goto 60                     if j>10 goto MoreThan10

  40 print " - Less than 10"             print " - Less Than 10"

  50 goto 70                             goto Continue

  60 print " - Greater than 10"         MoreThan10

  70 next                                print " - Greater Than 10"

                                        Continue

                                         next

Labels can be any string that starts with the letter A through Z. Label definitions must be the first thing on the line (flush left). They can be no more than 31 characters long, and can contain the letters A through Z, numbers 0 through 9, periods, and underscores. Case is not important - Metal treats "MoreThan10" the same as "morethan10" and "MORETHAN10".

Metal also imposes a "uniqueness" limit on labels. You cannot have two labels with the same name. For example:

       if x=1 goto Next

Next

       next:close:if not error goto Next

Next   <--- compiler will complain here

       return

The Compiler will complain where you defined the label "Next" the second time. If you need to use the same label name twice, see "local labels".

See also: local label, GLOBAL, .GLOBALALL, GOTO, GOSUB, RUN, RUNSUB

GLOBAL

Type: compiler directive

label GLOBAL

Entry GLOBAL

ErrorTrap GLOBAL

The GLOBAL directive is used right after the label definition. By making a label Global, you are letting other Metal programs enter into this point in the program. Normally, you only do this at places that you want to be entered into (like Entry or FromTrap or something similar), but you can make it whatever you want.

You cannot make local labels global.

The directive .GlobalAll has the same effect of putting GLOBAL after each label.

See also: label, .GLOBALALL, RUN, RUNSUB

]locallabel

Type: compiler directive

]L1

 goto ]Next

Local labels are just like standard labels, but you cannot make them Global (and thus cannot use a RunSub to enter into them from another program).

However, more than one local label can have the same name, provided they are not in the same set of labels.

Each time a "normal" label like Enter, MainLoop, TransferStats is defined, the Compiler starts a new "label set". From the point where the label is defined until another label is defined, you can use local labels.

Local labels all start with the right-square bracket ("]"). They are defined and used this way.

For example, the following is legal:

Main

       for j = 1 to 50

         if J>20 goto ]Next

       print J

]Next

       next

]Exit

       return

Somewhere

       goto ]Exit   <--- refers to ]Exit in Somewhere - down below

       print "Hi!"

]Exit

       goto Main

However, the following is not legal:

Main

       for j = 1 to 50

          if J>20 goto ]Next

       print J

]Next

       next

]Next  <--- duplicate label

       return

Somewhere

       goto ]Next   <--- there is no ]Next in Somewhere

       ...etc...

Local labels are powerful tools in programming, since you don't have to overload the Compiler (and your source) with labels like "MainJ1" and "MainLoop7".

See also: labels

$hexnum

Type: compiler directive

a=$abcd

print $f

if b=$02 goto bit1

Metal and the Compiler lets you put numbers in your source in three different ways: decimal (1, 19, 2001) and hexadecimal ($1, $13, $7D1) and binary (%0100).

Hexadecimal ("hex" for short) is base-16 counting, going from 0 to 9, with A being 10, B is 11, C is 12, etc, to F for 15.

Metal actually converts all numbers (decimal, hex, and binary) that you put in your source into hex numbers - the computer understands hex better than decimal.

Hex numbers are handy for some things, like file types and file lengths (a Super High-Res picture is $8000 bytes long, or 32768 bytes if you're not a programmer). For the most part, you won't need to use them, but if you do, they are there.

The Compiler can handle up to six hex digits in a row. After that, it gets a little confused, so be careful.

Note: using $hex in Metal is the same as 0xhex in C/C++.

See also: %bitnum, HEXSTR$, HEXVAL

%bitnum

Type: compiler directive

a=%10011101

if z b.and %101 goto bitson

Using bit numbers (base two, zeros and ones) is great for testing and setting single bit "flags" in numbers.

Like hex numbers, bit numbers are optional, and Metal can translate freely between all three sets.

See also: $hexnum, BITSTR$, BITVAL

varname

Type: compiler directive

varname{(array1{,array2})}

varname{(array1{,array2})}${[index{,length}]}

a=6

if you$="Fred" goto Hi

r=len(zy$)*usr9

One of the most important aspects of Basic languages is variables. Variables are the data sets of Basic, and are the most common idea to all languages. In fact, they actually originated in math formulas, like the ones you did for your homework:

          a = b - 2

          Solve for A if B is 7.

However, math usually never used any more than A through Z - sometimes with sub-scripts or super-scripts, but not often. Metal, on the other hand, allows you to use A through Z and then some.

Varables are considered variables when they meet the following rules:

1). The first character is a letter from A to Z.

2). The first few characters do not match a command word (like TO or FOR).

3). They contain only letters and numbers, if any.

4). If the last character is a dollar sign ("$"), then the variable is a string, and cannot handle numerical data. Instead, it can only handle text data.

So Metal allows you to have A, AB, ABC, ABCD, A9BZ, B00Z, NAME$, etc. This gives Metal the ability to have 2 * (26 * ((26+10+1) ^ 3)) or 2,633,956 different possible variables.

Another aspect of variables is the ability to use arrays. Arrays can be considered random-access variables with a common name. For example:

  a(1)    a(94)     a(1993)

These are all array variables with the common name of "a". However, they do not conflict with each other in any way. Thus, a(1) can hold 74, and a(94) can hold 832, and neither one will "collide" with each other.

Arrays have special properties. The most obvious one is that you can maniplulate them using a For-Next loop, like this:

       For x = 1 to 50

       print a(x)

       next

And this would print out the value of a(1), a(2), a(3), etc.

You can also achieve the same effect using Metal's ability to "bulk-handle" arrays for you:

       print a(1) to a(50)

The output here will be exactly the same as from the above example. And yes, it works for Input, also.

The Clear command can be used on an array to clear it out:

       clear a(1) to a(50)

Is the same as:

       for x = 1 to 50

       a(x)=0

       next

String arrays like a$(50) and user$(2) can be handled the same as any other string variable, and have the same properties as the numerical arrays that we used above.

By adding another array value, you can have what is called "two-dimensional" arrays (ones that can go both left and right and up and down). These type of arrays look like this:

       a(5,20)=7         a(x,y)=z

The "bulk-handle" ability of Metal doesn't work as you would expect for two-dimensional arrays: it only operates with the first array, and doesn't manipluate the second.

String variables under Metal can be accessed in an optional manner that is termed "intra-string" manipulation.

For example, let's say we wanted to get the third and forth character from a string:

          result$=mid$(a$,3,2)

We can, however, express this in a different way that is much faster:

          result$=a$[3,2]

The reason that it is faster is that Metal doesn't have to perform string stacking (which eats up memory and time) during the a$[x,y] calculation - all it does is a "block move" of part of the string.

Using the intra-string expression, you can change part of a string:

       a$[3,2]="Hi"

So if a$ was "Smile!", the above example would make it "SmHile!"

If the second part of the intra-string is missing, Metal interprets the code to mean that you wish to read or change the string from the given starting character to the end.

See also: _varname, localvariables

_varname

Type: compiler directive

_to=0

if _bit then return

By puting an underscore ("_") in front of a variable name, you can force the Compiler to use variable names that it normally would think were command words - like TO and BIT and FOR and NEXT.

See also: varname, localvariables

&localvariable

Type: compiler directive

&varnum{(array1{,array2})}

&varnum{(array1{,array2})}${[index{,length}]}

&1=7

&9$="You idiot!"

Local variables are a powerful feature of Metal, allowing the programmer to have separate variables for each routine, with the trade off in that these variables are very strange looking (&9$ for example).

Each time Metal runs the Gosub command, the current set of local variables become inactive, and the Gosub being executed has it's own private set of variables, which become discarded when that Gosub routine exits through the Return command.

Local variables are also used by the Gosub command to pass data into and out of routines without having to resort to pre-loading and post-reading variables (although these are valid programming technique).

When entering into a Gosub via the following:

       Gosub MyRoutine(a,b,c)

The local variable &0 will hold a value of 3, indicating that 3 values were passed to the routine: &1 will hold the value of the first one, &2 will hold the second, and &3 will hold the value of the third.

       Gosub MyOtherRoutine(a,b$,c)

&0 will also be 3 for this one, but &2 will hold garbage - in this case, &2$ will be holding the string value of the second expression, since it was a string.

You don't have to use real variables to pass data into the local variables - you can use expressions just as well:

       Gosub YetAnother(8,b$+","+date$)

Local variables are also used for passing data from Gosub routines back to the main caller. For example:

       Gosub Times10(5)(x)

Upon entry to the routine Times10, &1 will hold the value of 5. If the routine does the following:

       &1 = &1 * 10

This means that the local variable &1 will be multiplied by 10. When the routine exits, Metal sees that you wish to pass data back from local variables to the caller. Thus &1 is passed back into X, in this case making it equal 50.

When passing back data, you may mix numeric and string variables as long as that is all you use - you can not pass data back into expressions, since you are implicitly equating variables.

You don't have to pass anything to the routine if you don't want to:

       Gosub GetTimeOnline()(t)

Metal sees that this is passing an "empty set" to the routine (&0 will be zero, of course), but sees that you wish to pass back data to the caller. This is fine and very legal.

Local variables can be used in exactly the same fashion as real variables - you can assign them, clear them, use them with arrays, push them and pull them from the stack, etc.

Local variables names can go from &0 to &99, and &0$ to &99$, inclusive.

See also: varname, _varname
